Компьютерная математика. Символьный пакет Mathematica

Часть 2

Тема 3. Математический анализ в среде Mathematica. Многочлен Тейлора

Лаврова Ольга Анатольевна

БЕЛОРУССКИЙ ГОСУДАРСТВЕННЫЙ УНИВЕРСИТЕТ

ММФ, кафедра дифференциальных уравнений и системного анализа (ауд. 329)

Теоретические сведения

Ряд Тейлора -- это степенной ряд вида

$$f(x_0)+(x-x_0)\,f^{(1)}(x_0)+\frac{(x-x_0)^2\,f^{(2)}(x_0)}{2!}+\ldots+\frac{(x-x_0)^k\,f^{(k)}(x_0)}{k!}+\ldots=\sum_{k=0}^{\infty}\frac{(x-x_0)^k\,f^{(k)}(x_0)}{k!} \qquad (1)$$

где числовая функция $f(x)$ определена в некоторой окрестности точки $x = x_0$ и имеет в этой точке производные всех порядков.

Многочлен Тейлора степени n для функции $f(x)$, которая n раз дифференцируема при $x = x_0$, -- это частичная сумма ряда Тейлора

$$P_n(x)=\sum_{k=0}^{n}\frac{(x-x_0)^k\,f^{(k)}(x_0)}{k!}=f(x_0)+(x-x_0)\,f^{(1)}(x_0)+\frac{(x-x_0)^2\,f^{(2)}(x_0)}{2!}+\ldots+\frac{(x-x_0)^n\,f^{(n)}(x_0)}{n!}. \qquad (2)$$

Формула Тейлора -- это представление функции в виде ее многочлена Тейлора степени n $(n = 0, 1, 2, \ldots)$ и остаточного члена. Если действительная функция $f(x)$ имеет n производных в точке $x = x_0$, то ее формула Тейлора имеет вид

$$f(x) = P_n(x) + r_n(x), \qquad (3)$$

где остаточный член может быть записан в **форме Пеано** как

$$r_n(x) = o((x - x_0)^n)$$

при $x \to x_0$.

Многочлен Тейлора является многочленом наилучшего приближения функции $f(x)$ при $x \to x_0$ в том смысле, что

$$f(x) - P_n(x) = o((x - x_0)^n)$$

при $x \to x_0$.

Источник: Математическая энциклопедия / Гл. ред. И. М. Виноградов. -- М.: Советская энциклопедия. т. 5 Слу--Я. 1984.

Встроенная функция Series

Встроенная функция **Series[*f*, {*x*, *x*$_0$, *n*}]** представляет заданную функцию $f(x)$ в окрестности точки $x = x_0$ в виде суммы многочлена Тейлора заданной степени n и остаточного члена в форме Пеано

In[1]:= `?Series`

Series[f, {x, x_0, n}] generates a power series expansion for f about the point $x = x_0$ to order $(x - x_0)^n$.
Series[f, {x, x_0, n_x}, {y, y_0, n_y}, …] successively finds series expansions with respect to x, then y, etc. ≫

Возьмем произвольную функцию $y = f(x)$ и построим для нее формулу Тейлора в окрестности точки $x = a$ до степени $n = 3$

In[2]:= `Series[f[x], {x, a, 3}]`

Out[2]= $f[a] + f'[a]\,(x - a) + \frac{1}{2} f''[a]\,(x - a)^2 + \frac{1}{6} f^{(3)}[a]\,(x - a)^3 + O[x - a]^4$

Построение формулы Тейлора для заданной функции является примером символьных вычислений. Символьное выражение строится на основе производных от функции.

Возьмем функцию $y = \sin(x)$ и представим ее по формуле Тейлора в окрестности точки $x = 0$ до степени 10

In[3]:= `Series[Sin[x], {x, 0, 10}]`

Out[3]= $x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880} + O[x]^{11}$

Возьмем функцию $y = e^x$ и представим ее по формуле Тейлора в окрестности точки $x = 0$ до степени 10

In[4]:= `Series[Exp[x], {x, 0, 10}]`

Out[4]= $1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040} + \frac{x^8}{40\,320} + \frac{x^9}{362\,880} + \frac{x^{10}}{3\,628\,800} + O[x]^{11}$

Функция **Normal** позволяет получить многочлен Тейлора из формулы Тейлора, отбросив остаточный член

In[5]:=
```
Normal @ Series[Sin[x], {x, 0, 10}]
```

Out[5]= $x-\frac{x^3}{6}+\frac{x^5}{120}-\frac{x^7}{5040}+\frac{x^9}{362\,880}$

In[6]:=
```
expTaylor10 = Normal @ Series[Exp[x], {x, 0, 10}]
```

Out[6]= $1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120}+\frac{x^6}{720}+\frac{x^7}{5040}+\frac{x^8}{40\,320}+\frac{x^9}{362\,880}+\frac{x^{10}}{3\,628\,800}$

С помощью функции Manipulate построим интерактивный объект для демонстрации работы функции Series

In[7]:=
```
Manipulate[TableForm@{
   series = Normal@Series[f[x], {x, a, n}],
   Show[
    Plot[{f[x], series}, {x, a - 2, a + 2}, Filling → {1 → {2}}],
    Graphics[{PointSize[0.05], Red, Point@{a, f[a]}}]
   ]},
 {{f, Sin, "function"}, {Sin, Cos, Tan, Cot, Sec, Csc}},
 {{a, 0}, -2, 2, Appearance → "Labeled"},
 {n, 0, 100, 1, Appearance → "Labeled"}]
```

Out[7]=



Построение списка всех многочленов Тейлора до заданной степени

Способ 1. Использование функции Table

In[8]:=
```
Table[Normal @ Series[Sin[x], {x, 0, n}], {n, 1, 10}] // TableForm
```

Out[8]//TableForm=

x

x

$x - \frac{x^3}{6}$

$x - \frac{x^3}{6}$

$x - \frac{x^3}{6} + \frac{x^5}{120}$

$x - \frac{x^3}{6} + \frac{x^5}{120}$

$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040}$

$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040}$

$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880}$

$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880}$

При таком способе построения многочлены в списке могут повторяются

Способ 2. Выделение подвыражений с помощью Part

Построим многочлен Тейлора для функции Sin[x] 10-ой степени и для функции Exp[x] 10-ой степени

In[9]:=
```
sinTaylor10 = Normal @ Series[Sin[x], {x, 0, 10}]
```

Out[9]= $x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880}$

In[10]:=
```
expTaylor10 = Normal @ Series[Exp[x], {x, 0, 10}]
```

Out[10]= $1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040} + \frac{x^8}{40\,320} + \frac{x^9}{362\,880} + \frac{x^{10}}{3\,628\,800}$

Количество слагаемых многочлена Тейлора определим с помощью функции Length

In[11]:=
```
Length@sinTaylor10
```

Out[11]= 5

Использование функции **Part** в сочетании со **Span** позволяет выделять части исходного выражения. Например,

In[12]:=
```
sinTaylor10[[;; -1]]
```

Out[12]= $x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880}$

Применим функцию для построения многочленов Тейлора меньшей степени

In[13]:=
```
(listSinTaylor10 = Table[sinTaylor10[[;; n]], {n, Length@sinTaylor10}]) // TableForm
```

Out[13]//TableForm=

x

$x - \frac{x^3}{6}$

$x - \frac{x^3}{6} + \frac{x^5}{120}$

$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040}$

$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880}$

In[14]:=
```
(listExpTaylor10 = Table[expTaylor10[[;; n]], {n, Length@expTaylor10}]) // TableForm
```

Out[14]//TableForm=

1

$1 + x$

$1 + x + \frac{x^2}{2}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040} + \frac{x^8}{40\,320}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040} + \frac{x^8}{40\,320} + \frac{x^9}{362\,880}$

$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040} + \frac{x^8}{40\,320} + \frac{x^9}{362\,880} + \frac{x^{10}}{3\,628\,800}$

Разделение многочленов Тейлора по четным и нечетным степеням

По заданному списку многочленов необходимо построить список многочленов нечетной степени и список многочленов четной степени.

Одной из возможностей реализации является использование функции **Select**

In[15]:=
```
? Select
```

Select[*list*, *crit*] picks out all elements e_i of *list* for which *crit*[e_i] is True.
Select[*list*, *crit*, *n*] picks out the first *n* elements for which *crit*[e_i] is True.
Select[*crit*] represents an operator form of Select that can be applied to an expression. >>

Вторым аргументом функции Select нужно использовать функцию-предикат, которая определяет четность или нечетность старшей степени многочлена. Определим функцию-предикат как безымянную функцию с помощью встроенных функций **Exponent**, **OddQ**, **EvenQ**

In[16]:=
```
OddQ@Exponent[#, x] &
```

Out[16]=
```
OddQ[Exponent[#1, x]] &
```

In[17]:=
```
EvenQ@Exponent[#, x] &
```

Out[17]=
```
EvenQ[Exponent[#1, x]] &
```

In[18]:=
```
OddQ@Exponent[#, x] & /@ {x, x^2 + 1}
```

Out[18]=
```
{True, False}
```

In[19]:=
```
EvenQ@Exponent[#, x] & /@ {x, x^2 + 1}
```

Out[19]=
```
{False, True}
```

Строим списки с помощью функции Select

In[20]:=
```
Select[listSinTaylor10, OddQ@Exponent[#, x] &]
```

Out[20]=
$$\left\{x,\ x-\frac{x^3}{6},\ x-\frac{x^3}{6}+\frac{x^5}{120},\ x-\frac{x^3}{6}+\frac{x^5}{120}-\frac{x^7}{5040},\ x-\frac{x^3}{6}+\frac{x^5}{120}-\frac{x^7}{5040}+\frac{x^9}{362\,880}\right\}$$

In[21]:=
```
Select[listSinTaylor10, EvenQ@Exponent[#, x] &]
```
Out[21]= { }

Альтернативной возможностью реализации является использование функции **Cases** и **DeleteCases**

In[22]:=
```
? Cases
```

Cases[{e_1, e_2, …}, *pattern*] gives a list of the e_i that match the pattern.

Cases[{e_1, …}, *pattern* → *rhs*] gives a list of the values of *rhs* corresponding to the e_i that match the pattern.

Cases[*expr*, *pattern*, *levelspec*] gives a list of all parts of *expr* on levels specified by *levelspec* that match the pattern.

Cases[*expr*, *pattern* → *rhs*, *levelspec*] gives the values of *rhs* that match the pattern.

Cases[*expr*, *pattern*, *levelspec*, *n*] gives the first *n* parts in *expr* that match the pattern.

Cases[*pattern*] represents an operator form of Cases that can be applied to an expression. ≫

Вторым аргументом функции Cases нужно использовать образец, который определяет множество многочленов нечетной и четной степеней. Образец с условием можно определить двумя способами

In[23]:=
```
MatchQ[x, _? (OddQ[Exponent[#, x]] &)]
```
Out[23]= True

In[24]:=
```
MatchQ[x, pol_ /; EvenQ[Exponent[pol, x]]]
```
Out[24]= False

Строим список многочленов нечетной степени с помощью функции Cases

In[25]:=
```
oddListExpTaylor10 = Cases[listExpTaylor10, _? (OddQ[Exponent[#, x]] &)]
```
Out[25]= $\left\{1+x,\ 1+x+\frac{x^2}{2}+\frac{x^3}{6},\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120},\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120}+\frac{x^6}{720}+\frac{x^7}{5040},\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120}+\frac{x^6}{720}+\frac{x^7}{5040}+\frac{x^8}{40320}+\frac{x^9}{362880}\right\}$

Строим список многочленов четной степени с помощью функции **DeleteCases** удалением многочленов нечетной степени

In[26]:=
```
evenListExpTaylor10 = DeleteCases[listExpTaylor10, pol_ /; OddQ[Exponent[pol, x]]]
```
Out[26]= $\left\{1,\ 1+x+\frac{x^2}{2},\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24},\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120}+\frac{x^6}{720},\right.$
$\left. 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120}+\frac{x^6}{720}+\frac{x^7}{5040}+\frac{x^8}{40320},\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120}+\frac{x^6}{720}+\frac{x^7}{5040}+\frac{x^8}{40320}+\frac{x^9}{362880}+\frac{x^{10}}{3628800}\right\}$

Построение графиков многочленов Тейлора

Использование функции Plot

Построим графики многочленов Тейлора функции exp(*x*) нечетной степени. Обратите внимание на использование опций и функции **Directive** для установки нескольких значений опции.

In[27]:=
```
taylorOdd = Plot[oddListExpTaylor10, {x, -2, 2}, PlotStyle → Directive[Dashed, Green]]
```

Out[27]=



Построим графики многочленов Тейлора четной степени

In[28]:=
```
taylorEven = Plot[evenListExpTaylor10, {x, -2, 2}, PlotStyle → Directive[Dotted, Blue]]
```

Out[28]=



Объединим графики в одной графической области с помощью функции **Show**, добавим график функции exp(x) и точку, в окрестности которой строятся многочлены Тейлора

In[29]:=
```
Show[taylorOdd, taylorEven, Plot[Exp[x], {x, -2, 2}, PlotStyle → Red, PlotLegends → {Exp[x]}],
 Graphics[{PointSize[0.03], Point@{0, Exp[0]}}, Red], PlotRange → {-1, 3}]
```

Out[29]=



Использование функции ListPlot

Построим матричное представление многочленов Тейлора по их координатам

In[30]:=
```
matricesOdd = Table[{x, #}, {x, -2, 2, 0.1}] & /@ oddListExpTaylor10;
```

Например, первому многочлену в списке

In[31]:=
```
First@oddListExpTaylor10
```

Out[31]= $1 + x$

соответствует матрица

In[32]:=
```
First@matricesOdd // MatrixForm
```

Out[32]//MatrixForm=

$$\begin{pmatrix} -2. & -1. \\ -1.9 & -0.9 \\ -1.8 & -0.8 \\ -1.7 & -0.7 \\ -1.6 & -0.6 \\ -1.5 & -0.5 \\ -1.4 & -0.4 \\ -1.3 & -0.3 \\ -1.2 & -0.2 \\ -1.1 & -0.1 \\ -1. & 0. \\ -0.9 & 0.1 \\ -0.8 & 0.2 \\ -0.7 & 0.3 \\ -0.6 & 0.4 \\ -0.5 & 0.5 \\ -0.4 & 0.6 \\ -0.3 & 0.7 \\ -0.2 & 0.8 \\ -0.1 & 0.9 \\ 0. & 1. \\ 0.1 & 1.1 \\ 0.2 & 1.2 \\ 0.3 & 1.3 \\ 0.4 & 1.4 \\ 0.5 & 1.5 \\ 0.6 & 1.6 \\ 0.7 & 1.7 \\ 0.8 & 1.8 \\ 0.9 & 1.9 \\ 1. & 2. \\ 1.1 & 2.1 \\ 1.2 & 2.2 \\ 1.3 & 2.3 \\ 1.4 & 2.4 \\ 1.5 & 2.5 \\ 1.6 & 2.6 \\ 1.7 & 2.7 \\ 1.8 & 2.8 \\ 1.9 & 2.9 \\ 2. & 3. \end{pmatrix}$$

Построение графикой функций по их матричному представлению осуществляется с помощью функции **ListPlot**

In[33]:=
```
ListPlot[matricesOdd, PlotStyle → Directive[Dashed, Green]]
```

Out[33]=



Использование графических примитивов

Используя матричное представление функций по координатам точек, можно строить графики функций с помощью графического примитива **Line.**
Функция **Graphics** создает графический объект из графических примитивов.

In[34]:=
```
taylorOdd = Graphics[{Green, Line /@ matricesOdd}, Axes → True, PlotRange → {-1, 3}]
```

Out[34]=



Пользовательская функция построения многочлена Тейлора для явно заданной функции одной переменной

Многочлен Тейлора для функции $y = f(x)$, дифференцируемой n раз в точке x_0, имеет следующий вид

$$P_n(x) = \sum_{k=0}^{n} \frac{(x - x_0)^k f^{(k)}(x_0)}{k!}.$$

Необходимо написать глобальное правило преобразований, которое для заданной функции $f(x)$, точки x0 и степени n строит соответствующий многочлен Тейлора.

Определим формат для левой части глобального правила преобразований, аналогичный встроенной функции Series

```
In[35]:= ClearAll[series]

In[36]:= series[f_, {x_Symbol, x0_, n_Integer?NonNegative}]
Out[36]= series[f_, {x_Symbol, x0_, n_Integer?NonNegative}]

In[37]:= series::usage = "series[f,{x,x0,n}] строит многочлен Тейлора n-ой степени для функции f в окрестности точки x=x0";

In[38]:= ?series
```

series[f,{x,x0,n}] строит многочлен Тейлора n-ой степени для функции f в окрестности точки x=x0

Способ 1. Использование функций NestList и FoldList

Реализуем функцию **series** с использованием функций высших порядков **NestList** и **FoldList** по следующему алгоритму:

- с помощью NestList строим список многочленов вида $(x - x_0)^k$ по степеням k от 0 до n
- с помощью FoldList строим список коэффициентов вида $\frac{f^{(k)}(x_0)}{k!}$ при k от 0 до n
- перемножаем скалярно построенные списки.

In[39]:=
```
ClearAll[f, x, x0]
```

In[40]:=
```
n = 10;
```

Создадим список многочленов вида $(x - x_0)^n$ с помощью функции **NestList**

In[41]:=
```
?NestList
```

NestList[*f*, *expr*, *n*] gives a list of the results of applying *f* to *expr* 0 through *n* times. ≫

In[42]:=
```
NestList[f, 1, n]
```

Out[42]=
```
{1, f[1], f[f[1]], f[f[f[1]]], f[f[f[f[1]]]], f[f[f[f[f[1]]]]], f[f[f[f[f[f[1]]]]]], f[f[f[f[f[f[f[1]]]]]]],
 f[f[f[f[f[f[f[f[1]]]]]]]], f[f[f[f[f[f[f[f[f[1]]]]]]]]], f[f[f[f[f[f[f[f[f[f[1]]]]]]]]]]}
```

In[43]:=
```
NestList[(x - x₀) * # &, 1, n]
```

Out[43]= $\{1,\ x - x_0,\ (x - x_0)^2,\ (x - x_0)^3,\ (x - x_0)^4,\ (x - x_0)^5,\ (x - x_0)^6,\ (x - x_0)^7,\ (x - x_0)^8,\ (x - x_0)^9,\ (x - x_0)^{10}\}$

Создадим список коэффициентов многочлена Тейлора с помощью функции **FoldList**

In[44]:=
```
?FoldList
```

FoldList[*f*, *x*, {*a*, *b*, …}] gives {*x*, *f*[*x*, *a*], *f*[*f*[*x*, *a*], *b*], …}.
FoldList[*f*, {*a*, *b*, *c*, …}] gives {*a*, *f*[*a*, *b*], *f*[*f*[*a*, *b*], *c*], …}.
FoldList[*f*] represents an operator form of FoldList that can be applied to expressions. ≫

In[45]:= $\text{FoldList}\left[\frac{\text{D}[\#1, x_0]}{\#2}\ \&,\ f[x_0],\ \text{Range}[10]\right]$

Out[45]= $\left\{f[x_0],\ f'[x_0],\ \frac{1}{2} f''[x_0],\ \frac{1}{6} f^{(3)}[x_0],\ \frac{1}{24} f^{(4)}[x_0],\ \frac{1}{120} f^{(5)}[x_0],\ \frac{1}{720} f^{(6)}[x_0],\ \frac{f^{(7)}[x_0]}{5040},\ \frac{f^{(8)}[x_0]}{40\,320},\ \frac{f^{(9)}[x_0]}{362\,880},\ \frac{f^{(10)}[x_0]}{3\,628\,800}\right\}$

Перемножим скалярно построенные списки

In[46]:=
```
NestList[(x - a) * # &, 1, n].FoldList[D[#1, a]/#2 &, f[a], Range[10]]
```

Out[46]=
$$f[a] + (-a + x)\, f'[a] + \frac{1}{2}(-a + x)^2 f''[a] + \frac{1}{6}(-a + x)^3 f^{(3)}[a] + \frac{1}{24}(-a + x)^4 f^{(4)}[a] + \frac{1}{120}(-a + x)^5 f^{(5)}[a] + \frac{1}{720}(-a + x)^6 f^{(6)}[a] + \frac{(-a + x)^7 f^{(7)}[a]}{5040} + \frac{(-a + x)^8 f^{(8)}[a]}{40\,320} + \frac{(-a + x)^9 f^{(9)}[a]}{362\,880} + \frac{(-a + x)^{10} f^{(10)}[a]}{3\,628\,800}$$

Проверяем

In[47]:=
```
Normal @ Series[f[x], {x, a, 10}]
```

Out[47]=
$$f[a] + (-a + x)\, f'[a] + \frac{1}{2}(-a + x)^2 f''[a] + \frac{1}{6}(-a + x)^3 f^{(3)}[a] + \frac{1}{24}(-a + x)^4 f^{(4)}[a] + \frac{1}{120}(-a + x)^5 f^{(5)}[a] + \frac{1}{720}(-a + x)^6 f^{(6)}[a] + \frac{(-a + x)^7 f^{(7)}[a]}{5040} + \frac{(-a + x)^8 f^{(8)}[a]}{40\,320} + \frac{(-a + x)^9 f^{(9)}[a]}{362\,880} + \frac{(-a + x)^{10} f^{(10)}[a]}{3\,628\,800}$$

На основании реализованного алгоритма строим пользовательскую функцию

In[48]:=
```
series[f_, {x_Symbol, x0_, n_Integer?NonNegative}] := Block[{t},
  (NestList[(x - x0) * # &, 1, n].FoldList[D[#1, t]/#2 &, f[t], Range[n]]) /. t → x0]
```

При реализации функции использована локальная переменная *a*.

Протестируем работу построенной функции

In[49]:=
```
series[f, {x, c, 10}]
```

Out[49]=
$$f[c] + (-c + x)\, f'[c] + \frac{1}{2}(-c + x)^2 f''[c] + \frac{1}{6}(-c + x)^3 f^{(3)}[c] + \frac{1}{24}(-c + x)^4 f^{(4)}[c] + \frac{1}{120}(-c + x)^5 f^{(5)}[c] + \frac{1}{720}(-c + x)^6 f^{(6)}[c] + \frac{(-c + x)^7 f^{(7)}[c]}{5040} + \frac{(-c + x)^8 f^{(8)}[c]}{40\,320} + \frac{(-c + x)^9 f^{(9)}[c]}{362\,880} + \frac{(-c + x)^{10} f^{(10)}[c]}{3\,628\,800}$$

In[50]:=
```
{series[Sin, {y, a, 5}], Normal @Series[Sin[y], {y, a, 5}]} // TableForm
```

Out[50]//TableForm=
$$(-a + y)\, Cos[a] - \frac{1}{6}(-a + y)^3 Cos[a] + \frac{1}{120}(-a + y)^5 Cos[a] + Sin[a] - \frac{1}{2}(-a + y)^2 Sin[a] + \frac{1}{24}(-a + y)^4 Sin[a]$$
$$(-a + y)\, Cos[a] - \frac{1}{6}(-a + y)^3 Cos[a] + \frac{1}{120}(-a + y)^5 Cos[a] + Sin[a] - \frac{1}{2}(-a + y)^2 Sin[a] + \frac{1}{24}(-a + y)^4 Sin[a]$$

In[51]:=
```
{series[Sin[#] + # &, {x, a, 5}], Normal @Series[Sin[x] + x, {x, a, 5}]} // TableForm
```

Out[51]//TableForm=

$$a - \frac{1}{6}(-a+x)^3 \operatorname{Cos}[a] + \frac{1}{120}(-a+x)^5 \operatorname{Cos}[a] + (-a+x)\left(1+\operatorname{Cos}[a]\right) + \operatorname{Sin}[a] - \frac{1}{2}(-a+x)^2 \operatorname{Sin}[a] + \frac{1}{24}(-a+x)^4 \operatorname{Sin}[a]$$

$$a - \frac{1}{6}(-a+x)^3 \operatorname{Cos}[a] + \frac{1}{120}(-a+x)^5 \operatorname{Cos}[a] + (-a+x)\left(1+\operatorname{Cos}[a]\right) + \operatorname{Sin}[a] - \frac{1}{2}(-a+x)^2 \operatorname{Sin}[a] + \frac{1}{24}(-a+x)^4 \operatorname{Sin}[a]$$

Способ 2. Использование функций Map и Apply

Реализуем функцию **series** с использованием функций высших порядков **Map** и **Apply** по следующему алгоритму:

- с помощью Map строим список многочленов вида $\frac{f^{(k)}(x_0)}{k!}(x-x_0)^k$ по степеням k от 0 до n
- с помощью Apply заменяем голову выражения List на Plus.

Данный подход был использован при реализации задачи о делении многочлена на многочлен.

In[52]:=
```
list = (1/#! D[f[x0], {x0, #}] * (x - x0)^#) & /@ Range[0, n]
```

Out[52]=

$$\left\{f[x0],\ (x-x0)\,f'[x0],\ \frac{1}{2}(x-x0)^2 f''[x0],\ \frac{1}{6}(x-x0)^3 f^{(3)}[x0],\ \frac{1}{24}(x-x0)^4 f^{(4)}[x0],\ \frac{1}{120}(x-x0)^5 f^{(5)}[x0],\ \frac{1}{720}(x-x0)^6 f^{(6)}[x0],\ \frac{(x-x0)^7 f^{(7)}[x0]}{5040},\ \frac{(x-x0)^8 f^{(8)}[x0]}{40\,320},\ \frac{(x-x0)^9 f^{(9)}[x0]}{362\,880},\ \frac{(x-x0)^{10} f^{(10)}[x0]}{3\,628\,800}\right\}$$

In[53]:=
```
Plus @@ list
```

Out[53]=

$$f[x0] + (x-x0)\,f'[x0] + \frac{1}{2}(x-x0)^2 f''[x0] + \frac{1}{6}(x-x0)^3 f^{(3)}[x0] + \frac{1}{24}(x-x0)^4 f^{(4)}[x0] + \frac{1}{120}(x-x0)^5 f^{(5)}[x0] + \frac{1}{720}(x-x0)^6 f^{(6)}[x0] + \frac{(x-x0)^7 f^{(7)}[x0]}{5040} + \frac{(x-x0)^8 f^{(8)}[x0]}{40\,320} + \frac{(x-x0)^9 f^{(9)}[x0]}{362\,880} + \frac{(x-x0)^{10} f^{(10)}[x0]}{3\,628\,800}$$

На основании реализованного алгоритма строим пользовательскую функцию

In[54]:=
```
series[f_, {x_Symbol, x0_, n_Integer?NonNegative}] := Block[{a},
  (Plus @@ ((1/#! D[f[a], {a, #}] * (x - a)^#) & /@ Range[0, n])) /. a → x0]
```

При реализации функции использована локальная переменная *a*.

Протестируем работу построенной функции

In[55]:=
```
{series[Exp[#] + # &, {x, a, 5}], Normal@Series[Exp[x] + x, {x, a, 5}]} // TableForm
```

Out[55]//TableForm=

$$a + e^a + (1 + e^a)(-a + x) + \frac{1}{2} e^a (-a + x)^2 + \frac{1}{6} e^a (-a + x)^3 + \frac{1}{24} e^a (-a + x)^4 + \frac{1}{120} e^a (-a + x)^5$$
$$a + e^a + (1 + e^a)(-a + x) + \frac{1}{2} e^a (-a + x)^2 + \frac{1}{6} e^a (-a + x)^3 + \frac{1}{24} e^a (-a + x)^4 + \frac{1}{120} e^a (-a + x)^5$$

Пользовательская функция построения многочленов Тейлора для параметрической функции

Пусть функция задана параметрически функциями $x(t)$ и $y(t)$.

Тогда формула многочлена Тейлора для явно заданной функции $y = f(x)$, дифференцируемой n раз в окрестности точки $x_0 = x(t_0)$

$$P_n(x) = \sum_{k=0}^{n} \frac{(x - x_0)^k f^{(k)}(x_0)}{k!}$$

записывается в окрестности точки $t = t_0$ с учетом параметрического задания функции

$$P_n(x) = \sum_{k=0}^{n} \frac{(x - x(t_0))^k \frac{d^k y(t_0)}{dx^k}}{k!}$$

Основное изменение связано с необходимостью вычисления производной от функции, заданной параметрически

Дифференцирование параметрической функции

Сначала определим правила вычисления нулевой производной от заданной параметрической функции и первой производной $\frac{dy}{dx} = \frac{\frac{dy}{dt}}{\frac{dx}{dt}}$

In[56]:=
```
dParametric[{xt_, yt_}, {t_Symbol, 0}] := yt // FullSimplify
```

In[57]:=
```
dParametric[{xt_, yt_}, {t_Symbol, 1}] := D[yt, t] / D[xt, t] // FullSimplify
```

Производные второго порядка и выше определяются формулой $\frac{d^n y}{dx^n} = \frac{\frac{d}{dt}\frac{d^{n-1}y}{dx^{n-1}}}{\frac{dx}{dt}}$ и могут быть

реализованы **рекурсивно**

In[58]:=
```
dParametric[{xt_, yt_}, {t_Symbol, n_Integer?Positive}] := D[dParametric[{xt, yt}, {t, n - 1}], t] / D[xt, t] // FullSimplify
```

Тестируем

In[59]:=
```
dParametric[{t, t^3}, {t, 2}]
```
Out[59]= $6\,t$

In[60]:=
```
dParametric[{t + Log[Cos[t]], t - Log[Sin[t]]}, {t, 2}]
```
Out[60]= $\frac{\mathrm{Csc}[t]^2}{1 - \mathrm{Tan}[t]}$

Источник: Л.Л. Голубева, А.Э. Малевич, Н.Л. Щеглова. Компьютерная математика. Символьный пакет Mathematica: лабораторный практикум -- Минск.: БГУ, 2012. Стр. 28--31.

Реализация пользовательской функции

In[61]:=
```
ClearAll[seriesParametric]
```

In[62]:=
```
seriesParametric::usage = "seriesParametric[{xt,yt},{t,t0,n}] строит
    многочлен Тейлора n-ой степени для параметрической функции {xt,yt} в окрестности точки t=t0";
```

In[63]:=
```
? seriesParametric
```

seriesParametric[{xt,yt},{t,t0,n}] строит многочлен Тейлора n-ой степени для параметрической функции {xt,yt} в окрестности точки t=t0

Определим формат для левой части глобального правила преобразований

In[64]:=
```
seriesParametric[{xt_, yt_}, {t_Symbol, t0_, n_Integer?NonNegative}]
```
Out[64]=
```
seriesParametric[{xt_, yt_}, {t_Symbol, t0_, n_Integer?NonNegative}]
```

За основу возьмем реализацию функции series с использованием функций **Map** и **Apply**

In[65]:=
```
series[f_, {x_Symbol, x0_, n_Integer?NonNegative}] := Block[{a},
  (Plus @@ ((1/#! D[f[a], {a, #}] * (x - a)^#) & /@ Range[0, n])) /. a -> x0]
```

Перечислим корректировки, которые необходимо осуществить при построении многочлена Тейлора для параметрической функции:

- функцию *f* заменяем на список двух выражений для представления параметрической функции {xt,yt};
- переменную x0 заменяем на t0;

- функцию дифференцирования *D*[*f*[*a*], {*a*, #}] заменяем на функцию dParametric[{xt, yt}, {*t*, #}];
- определяем значение переменной *a* как выражение xt, вычисленное при $t = \text{t0}$;
- подстановку $a \to \text{x0}$ заменяем на подстановку $t \to \text{t0}$.

Внесем корректировки

In[66]:=
```
seriesParametric[{xt_, yt_}, {t_Symbol, t0_, n_Integer?NonNegative}] :=
 (Plus @@ ((1/#! dParametric[{xt, yt}, {t, #}] * (x - xt)^#) & /@ Range[0, n])) /. t → t0
```

Функция для построения многочлена Тейлора определена полностью

Тестируем

In[67]:=
```
{xt = t, yt = t^3};
```

In[68]:=
```
{seriesParametric[{xt, yt}, {t, 1, 2}], series[#^3 &, {x, xt /. t → 1, 2}]} // TableForm
```

Out[68]//TableForm=

$1 + 3 (-1 + x) + 3 (-1 + x)^2$

$1 + 3 (-1 + x) + 3 (-1 + x)^2$

С помощью функции Manipulate динамически визуализируем результат работы пользовательской функции seriesParametric

In[69]:=
```
Manipulate[TableForm@
  {seriesParametric[{xt, yt}, {t, t0, n}],
    Show[
     ParametricPlot[{xt, yt}, {t, -10, 10}, PlotRange → {-5, 5}],
     Graphics[{PointSize[0.04], Red, Point@{xt, yt} /. t → t0}],
     Plot[seriesParametric[{xt, yt}, {t, t0, n}], {x, -5, 5}, PlotStyle → Green]]
   },
  {{t0, 0}, -2, 2, Appearance -> "Labeled"},
  {n, 2, 100, 1, Appearance -> "Labeled"}]
```

Out[69]=



Важные понятия

- Встроенная функция **Series[*f*, {*x*, *x*$_0$, *n*}]** строит формулу Тейлора для заданной функции $f(x)$ в окрестности точки $x = x_0$ в виде многочлена Тейлора заданной степени n и остаточного членf в форме Пеано.
- Характерные черты функционального стиля программирования (безымянные функции, функции высшего порядка, отсутствие циклов, использование списков).
- Построение графиков функций можно осуществлять с помощью графических функций (например, **Plot** и **ListPlot**) или графического примитива **Line**.
    - Функция **Graphics** создает графический объект из графических примитивов.
    - Функция **Show** объединяет графические объекты в одной графической области.